

CS:5810

Formal Methods in Software Engineering

Introduction to Alloy

Part 2

*Copyright 2001-18, Matt Dwyer, John Hatcliff, Rod Howell, Laurence Pilard, and Cesare Tinelli.
Created by Cesare Tinelli and Laurence Pilard at the University of Iowa from notes originally developed by Matt Dwyer,
John Hatcliff, Rod Howell at Kansas State University. These notes are copyrighted materials and may not be used in other
course settings outside of the University of Iowa in their current form or modified form without the express written
permission of one of the copyright holders. During this course, students are prohibited from selling notes to or being paid
for taking notes by any person or commercial firm without the express written permission of one of the copyright holders.*

Alloys Constraints

- Signatures and fields resp. define classes (of atoms) and relations between them
- Alloy models can be refined further by adding **formulas** expressing additional constraints over those classes and relations
- Several operators are available to express both logical and relational constraints

Logical Operators

The usual logical operators are available, often in two forms

- **not** **!** (Boolean) negation
- **and** **&&** conjunction
- **or** **||** disjunction
- **implies** **=>** implication
- **else** alternative
- **<=>** equivalence

Quantifiers

Alloy includes a rich collection of quantifiers

all $x: S \mid F$ F holds for **every** x in S

some $x: S \mid F$ F holds for **some** x in S

no $x: S \mid F$ F holds for **no** x in S

1one $x: S \mid F$ F holds for **at most one** x in S

one $x: S \mid F$ F holds for **exactly one** x in S

Predefined Sets in Alloy

There are three predefined set constants:

- **none** : empty set
- **univ** : universal set
- **ident** : identity relation

Example. For a model instance with just:

`Man = {(M0),(M1),(M2)}`

`Woman = {(W0),(W1)}`

the constants have the values

`none = {}`

`univ = {(M0),(M1),(M2),(W0),(W1)}`

`ident = {(M0,M0),(M1,M1),(M2,M2),(W0,W0),(W1,W1)}`

Everything is a Set in Alloy

- There are **no scalars**
 - We never speak directly about elements (or tuples) of relations
 - Instead, we can use **singleton** relations:

one sig Matt **extends** Person

- Quantified variables **always** denote singleton relations:

all $x : S \mid \dots x \dots$

$x = \{t\}$ for some element t of S

Set Operators

| | |
|-------------|--------------|
| + | union |
| & | intersection |
| - | difference |
| in | subset |
| = | equality |
| \neq | disequality |

Example. Married men:

$\text{Married} \ \& \ \text{Man}$

Relational Operators

| | |
|---------------|------------------------------|
| \rightarrow | arrow (cross product) |
| \sim | transpose |
| \cdot | dot join |
| $[\]$ | box join |
| \wedge | transitive closure |
| $*$ | reflexive-transitive closure |
| $\langle :$ | domain restriction |
| $: \rangle$ | image restriction |
| $++$ | override |

Arrow Product

$p \rightarrow q$

- p and q are two relations
- $p \rightarrow q$ is the relation you get by taking every combination of a tuple from p and a tuple from q and concatenating them (same as *flat* cross product)

Examples.

Name = {(N0),(N1)}

Addr = {(D0),(D1)}

Book = {(B0)}

Name \rightarrow Addr = {(N0,D0),(N0,D1),(N1,D0),(N1,D1)}

Book \rightarrow Name \rightarrow Addr =

{(B0,N0,D0),(B0,N0,D1),(B0,N1,D0),(B0,N1,D1)}

Transpose

$\sim p$

take the mirror image of the relation p ,
i.e., reverse the order of atoms in each tuple

Example:

- $p = \{(a_0, a_1, a_2, a_3), (b_0, b_1, b_2, b_3)\}$
- $\sim p = \{(a_3, a_2, a_1, a_0), (b_3, b_2, b_1, b_0)\}$

How would you use \sim to express the parents relation ?

$\sim children$

Relational Composition (Join)

$p \cdot q$

- p and q are two relations that are **not both unary**
- $p \cdot q$ is the relation you get by taking every combination of a tuple from p and a tuple from q and adding their join, if it exists

How to join tuples ?

- What is the join of these two tuples ?
 - (a_1, \dots, a_m)
 - (b_1, \dots, b_n)

If $a_m \neq b_1$ then the join is undefined

If $a_m = b_1$ then it is: $(a_1, \dots, a_{m-1}, b_2, \dots, b_n)$

Examples.

- $(a,b).(a,c,d)$ undefined
- $(a,b).(b,c,d) = (a,c,d)$
- What about $(a) . (a)$? Not defined !

$t_1.t_2$ is not defined if t_1 and t_2 are **both** unary tuples

Examples

to maps a message to the name it should be sent to


address maps names to addresses

to = $\{(M0, N0), (M0, N2), (M1, N2), (M2, N3)\}$

address = $\{(N0, D0), (N0, D1), (N1, D1), (N2, D3)\}$

to.address maps a message to the addresses it should be sent to

to.address = $\{(M0, D0), (M0, D1), (M0, D3), (M1, D3)\}$



→ **to**
---→ **address**
—→ **to.address**

Exercises

What's the result of these join applications?

- $\{(a,b)\}.\{(c)\}$
- $\{(a)\}.\{(a,b)\}$
- $\{(a,b)\}.\{(b)\}$
- $\{(a)\}.\{(a,b,c)\}$
- $\{(a,b,c)\}.\{(c),(c,d),(b,c)\}$
- $\{(a,b)\}.\{(a,b,c)\}$
- $\{(a,b,c,d)\}.\{(d,e,f),(d,a)\}$
- $\{(a)\}.\{(b)\}$

Exercises

- Given a relation addr of arity 4 that contains the tuple $b \rightarrow n \rightarrow a \rightarrow t$ when book b maps name n to address a at time t , and given a specific book B and a time T :
 - $\text{addr} = \{(B0, N0, D0, T0), (B0, N0, D1, T1), (B0, N1, D2, T0), (B0, N1, D2, T1), (B1, N2, D3, T0), (B1, N2, D4, T1)\}$
 - $T = \{(T1)\}$ $B = \{(B0)\}$

The expression $B . \text{addr} . T$ is the name-address mapping of book B at time T . What is the value of $B . \text{addr} . T$?

- When p is a binary relation and q is a ternary relation, what is the arity of the relation $p.q$?
- Join is not associative, why ?
(i.e., $(p.q).r$ and $p.(q.r)$ are not always equivalent)

Example: Family Structure

```
abstract sig Person {  
  children: set Person,  
  siblings: set Person  
}
```

```
sig Man, Woman extends Person {}
```

```
sig Married in Person {  
  spouse: one Married  
}
```


Example: Family Structure

- How would you use join to find Matt's children or grandchildren ?
 - `matt.children` // Matt's children
 - `matt.children.children` // Matt's grandchildren
- What if we want to find Matt's descendants?

Example: Family Structure

Every married man (woman) has a wife (husband)

```
all p: Married |  
  (p in Man => p.spouse in woman)  
and  
  (p in woman => p.spouse in Man)
```

A spouse can't be a sibling

```
no p: Married |  
  p.spouse in p.siblings
```

Box Join

$p[q]$

- Semantically identical to dot join, but takes its arguments in different order

$$p[q] \equiv q.p$$


Example: Matt's children or grandchildren ?

- `children[matt]` // Matt's children
- `children.children[matt]` // Matt's grandchildren
- `children[children[matt]]` // Matt's grandchildren

Transitive Closure

$\wedge r$

- Intuitively, the transitive closure of a relation $r : S \times S$ is what you get when you keep navigating through r until you can't go any farther



$(S0, S1)$
 $(S1, S2)$
 $(S2, S3)$
 $(S4, S7)$
 $(S0, S2)$
 $(S0, S3)$
 $(S1, S3)$

$\wedge r$

$$\wedge r = r + r.r + r.r.r + \dots$$

Example: Family Structure

- What if we want to find Matt's ancestors or descendants ?
 - `matt.^children` // Matt's descendants
 - `matt.^(~children)` // Matt's ancestors
- How would you express the constraint “*No person can be their own ancestor*”

`no p: Person | p in p.^(~children)`

Reflexive-transitive closure

- $*r = \wedge r + \text{iden}$

(S0,S1)
(S1,S2)
(S2,S3)
(S4,S7)

r

(S0,S1)
(S1,S2)
(S2,S3)
(S4,S7)
(S0,S2)
(S0,S3)
(S1,S3)
(S0,S0)
(S1,S1)
(S2,S2)
(S3,S3)
(S4,S4)
(S7,S7)

$\wedge r$

$*r$

iden

Domain and Image Restrictions

The restriction operators are used to **filter** relations to a given domain or image

If S is a set and r is a relation then

- $S \prec r$ contains tuples of r **starting** with an element in S
- $r \succ S$ contains tuples of r **ending** with an element in S

Examples.

```
Man = {(M0),(M1),(M2),(M3)}
```

```
Woman = {(W0),(W1)}
```

```
children = {(M0,M1),(M0,M2),(M3,W0),(W1,M1)}
```

```
// father-child
```

```
Man  $\prec$  children = {(M0,M1),(M0,M2),(M3,W0)}
```

```
// parent-son
```

```
children  $\succ$  Man = {(M0,M1),(M0,M2),(W1,M1)}
```

Override


$p \ ++ \ q$

- p and q are two relations of **arity two or more**
- the result is like the union between p and q except that tuples of q can replace tuples of p ; any tuple in p that matches a tuple in q starting with the same element is dropped
- $p \ ++ \ q = p - (\text{domain}(q) <: p) + q$

Example

- $\text{oldAddr} = \{(N0,D0),(N1,D1),(N1,D2)\}$
- $\text{newAddr} = \{(N1,D4),(N3,D3)\}$
- $\text{oldAddr} \ ++ \ \text{newAddr} = \{(N0,D0),(N1,D4),(N3,D3)\}$

Operator Precedence



Example: Family Structure

How would you express the constraint “*No person can have more than one father and mother*” ?

Example: Family Structure

How would you express the constraint “*No person can have more than one father and mother*” ?

```
all p: Person |  
  (lone (children.p & Man)) and  
  (lone (children.p & woman))
```

Equivalently:

```
all p: Person |  
  (lone (Man <: children).p) and  
  (lone (woman <: children).p)
```

This is an example of a negative constraint that is easier to state positively (to make use of the **lone** operator)

Set Comprehension

$\{ x : S \mid F \}$

- the set of values drawn from set S for which F holds

How would use the comprehension notation to specify the set of people that have the same parents as Matt?

(assuming `Person` has a `parents` field)

Set Comprehension

$\{ x : S \mid F \}$

- the set of values drawn from set S for which F holds

How would use the comprehension notation to specify the set of people that have the same parents as Matt?

$\{ q : \text{Person} \mid q.\text{parents} = \text{matt}.\text{parents} \}$

(assuming Person has a parents field)

Example: Family Structure

How would you express the constraint *“A person P ’s siblings are those people, other than P , with the same parents as P ”*

Example: Family Structure

How would you express the constraint *“A person P’s siblings are those people, other than P, with the same parents as P”*

```
all p: Person |  
  p.siblings =  
    {q: Person | p.parents = q.parents} - p
```

Let

You can factor expressions out:

$\text{let } x = e \mid A$

- Each occurrence of the variable x will be replaced by the expression e in A

Example: *Each married man (woman) has a wife (husband)*

```
all p: Married |  
  let q = p.spouse |  
    (p in Man => q in Woman) and  
    (p in Woman => q in Man)
```


Facts

Additional constraints on signatures and fields are expressed in Alloy as **facts**

```
fact Name {  
    F1  
    F2  
    ...  
}
```

AA looks for instances of a model that also satisfy **all** of its fact constraints

Example Facts

-- No person can be their own ancestor

-- At most one father and mother

-- a persons's siblings are other persons with the same parents

Example Facts

-- No person can be their own ancestor

```
fact selfAncestor {  
  no p: Person | p in p.^parents  
}
```

-- At most one father and mother

```
fact loneParents {  
  all p: Person | lone (p.parents & Man) and  
                   lone (p.parents & woman)  
}
```

-- a persons's siblings are other persons with the same parents

```
fact siblingsDefinition {  
  all p: Person |  
    p.siblings = {q: Person | p.parents = q.parents} - p  
}
```

Example Facts

```
fact social {  
  -- Every married man (woman) has a wife (husband)  
  all p: Married |  
    let s = p.spouse |  
      (p in Man => s in Woman) and  
      (p in Woman => s in Man)  
  
  -- A spouse can't be a sibling  
  no p: Married | p.spouse in p.siblings  
  
  -- A person can't be married to a blood relative  
  no p: Married |  
    some (p.*parents & (p.spouse).*parents)  
}
```

Run Command

- Used to ask AA to generate an instance of the model
- May include **conditions**
 - Used to guide AA to pick model instances with certain characteristics
 - E.g., force certain **sets and relations** to be non-empty
 - In this case, not part of the “true” specification

Run Command

- To analyze a model, you add a **run** command and instruct AA to execute it.
 - the **run** command
 - tells the tool to search for an **instance** of the model
 - you may also give a **scope** to signatures
 - bounds the size** of instances that will be considered
- AA **executes only the first run** command in a file

Scope

- Limits the size of instances considered to make instance finding feasible
- Represents the maximum number of elements in a top-level signature
- Default value = 3 for each top-level signature

Run Conditions

- We can use **condition schemas** to encode *realism constraints* to e.g.,
 - Force generated models to include at least one married person, or one married man, etc.

Run Example

Family Structure:

```
-- The simplest run command
-- The scope of every signature is 3
run {}

-- The scope of every signature is 5
run {} for 5

-- With conditions forcing each set to be populated
-- Setting the scope to 2
run {some Man && some Woman && some Married} for 2

-- Other scenarios
run {some Woman && no Man} for 7
run {some Man && some Married && no Woman}
```

Exercises

- Load family-2.a1s
- Execute it
- Analyze the metamodel
- Look at the generated instance
- Does it look correct?
- What if anything would you change about it?

Empty Signatures

- The analyzer's algorithms prefer smaller instances
 - Often it produces empty signatures or otherwise trivial instances
 - It is useful to know that these instances satisfy the constraints (since you may not want them)
- Usually, they do not illustrate the interesting behaviors that are possible

Exercises

- Load family-3.a1s
- Execute it
- Look at the generated instance
- Does it look correct?
- How can you produce
 - two married couples?
 - a non empty married relation and a non-empty siblings relation ?

Assertions

- Often we believe that our model **entails** certain **constraints** that are not directly expressed
 - e.g., `some A && (A in B)` entails `some B`
- We can define these constraints as **assertions** and ask the analyzer to check if they hold
 - e.g., `assert myAssertion { some B }`
`check myAssertion`

Assertions

- If the constraint in an assertion does not hold, the analyzer will produce a **counterexample instance**
- If you expect the constraint to hold but it does not, you can either
 - make it into a fact, or
 - refine your model until the assertion holds

Assertions

- No person has a parent that is also a sibling

```
assert a1 { all p: Person |  
            no p.parents & p.siblings }
```

- A person's siblings are his/her siblings' siblings

```
assert a2 { all p: Person |  
            p.siblings = p.siblings.siblings }
```

- No person shares a common ancestor with his/her spouse (i.e., spouse isn't related by blood)

```
assert a3 { no p: Married |  
            some (p.^parents & p.spouse.^parents) }
```

Assertion Scopes

- You can specify a scope explicitly for any signature, but:
 - If a signature has been given a bound
 - Then the bound of **its supersignature** or **any other extension of the same supersignature** can be determined

Example Scope

```
abstract sig Object {}  
sig Directory extends Object {}  
sig File extend Object {}  
sig Alias in File {}
```

We consider an assertion A

- **well-formed:**
 - check A for 5 Object
 - check A for 4 Directory, 3 File
 - check A for 5 Object, 3 Directory
 - check A for 3 Directory, 3 Alias, 5 File
- **ill-formed** because it leaves the bound of `File` unspecified
 - check A for 3 Directory, 3 Alias

Example Scope

```
abstract sig Object {}  
sig Directory extends Object {}  
sig File extends Object {}  
sig Alias in File {}
```

- `check A for 5 [or] run {} for 5`
places a bound of 5 on each top-level signature (in this case just `Object`)
- `check A for 5 but 3 Directory`
additionally places a bound of 3 on `Directory`, and a bound of 2 on `File` by implication
- `check A for exactly 3 Directory, exactly 3 Alias, 5 File`
limits `File` to at most 5 tuples, but requires that `Directory` and `Alias` have exactly 3 tuples each

Size Determination

Size determined in a signature declaration has priority on size determined in scope

Example:

```
abstract sig Color {}  
one sig red, yellow, green extends color {}  
sig Pixel {color: one Color}
```

check A for 2

limits the signature `Pixel` to 2 elements, but assigns a size of exactly 3 to `Color`

Exercises

- Load family-4.a1s
- Execute it
- Look at the generated counter-examples
- Why is SiblingsSibling false?
- Why is NoIncest false?

Problems with Assertions

Analyzing SiblingSiblings ...

Scopes: Person(3)

Counterexample found:

Person = {M,W0,W1}

Man = {M}

Woman = {W0,W1}

Married = {M,W1}

children = {(W0,W1)}

siblings = {(M,W0),(W0,M)}

spouse = {(M,W1),(W1,M)}

M.siblings = {W0}

M.siblings.siblings = {M}

Problems with Assertions

Analyzing NoIncest ...

Scopes: Person(3)

Counterexample found:

Person = {M0,M1,W}

Man = {M0,M1}

Woman = {W}

Married = {M1,W}

children = {(M0,W),(W,M1)}

siblings = {}

spouse = {(M1,W),(W,M1)}

(M0 is an Ancestor of M1
and
M0 is an ancestor of W)
and
M1 and W are married

Exercises

- Fix the specification in `family-4.a1s`
 - If the model is underconstrained, add appropriate constraints
 - If the assertion is not correct, modify it
- Demonstrate that your fixes yield no counter-examples
 - Does varying the scope make a difference?
 - Does this mean that the assertions hold for all models?

Functions and Predicates

Parametrized macros for terms and formulas

- Can be named and reused in different contexts (facts, assertions and conditions of run)
- Can have zero or more parameters
- Used to factor out common patterns

Functions are good for:

- **set expressions** you want to reuse in different contexts

Predicates are good for:

- **formulas** you want to reuse in different contexts

Functions

A named **set expression**, with zero or more parameters

Examples:

- The sisters function

```
fun sisters [p: Person] : Woman {  
    {w: Woman | w in p.siblings} }
```

- The parents relation

```
fun parents [] : Person -> Person {~children}
```

- Used in a formula

```
all p: Person | not (p in p.^parents or  
    p in sisters[p])
```

Predicates

A named **formula**, with zero or more parameters

Predicates are **not** included when analyzing other schemas (e.g., facts or assertions) unless they are applied to actual arguments in the schemas being analyzed

Example:

- Two persons are blood relatives iff they have a common ancestor

```
pred BloodRelated [p: Person, q: Person] {  
  some (p.*parents & q.*parents)  
}
```

- A person can't be married to a blood relative

```
no p: Married | BloodRelated[p, p.spouse]
```

Predicate or Fact ?

- Predicates are (parametrized) **definitions** of constraints
- Facts are **assumed** constraints
- **Note:** You can package constraints as predicates and then use those predicates in facts

```
pred IsSingle[p: Person] { not (p in Married) }  
pred IsFather[p: Man] { some p.children }
```

```
fact { some q: Man | IsSingle[q] && IsFather[q] }
```

Exercises

- Define a **predicate** that characterizes the notion of “in-law” for the family example
- Write a **fact** stating that a person is an in-law of their in-laws
- Add these to the family example and **run** it through AA
- Can you express this same notion in another way in the Alloy model?
 - Do so and run it through AA
 - Which approach is better? Why?

Exercises

- Add an **assertion** stating that a person has no married in-laws
- What is the minimum **scope** for set Person for which ACA can find a counterexample?
- How would you use ACA to prove that your answer is truly the minimum scope?
- prove it!

Acknowledgements

The family structure example is based on an example by Daniel Jackson distributed with the Alloy Analyzer